BIOST P8110: Applied Regression II

Lecture Note 14 - Time-Dependent Covariates in Cox Models

Qixuan Chen

Department of Biostatistics

Columbia University

This lecture's big ideas

1. Cox models with time-dependent covariates
2. How to fit the models using PROC PHREG in SAS

Data Example

NAME:
STAN: Standford Heart Transplant Patients (Crowley and Hu; 1977)

SIZE:
103 observations

SOURCE:
The sample consisted of 103 cardiac patients who were enrolled in the transplantation program between 1967 and 1974. After enrollment, patients waited varying lengths of time until a suitable donor heart was found. Patients were followed until death or until the termination date of April 1, 1974. Of the 69 transplant recipients, only 24 were still alive at termination.

LIST OF VARIABLES:

Variables	Name	Description
1	DOB	is the date of birth
2	DOA	is the date of acceptance into the program
3	DOT	is the date of transplant
4	DLS	is the date last seen (death date or censoring date)
5	DEAD	status at last seen (1=dead; 0=otherwise)
6	SURG	had open-heart surgery before DOA (1=yes; 0=no)

1. Read in the data set and create variables needed for analysis

```
/*Read in the data set*/
libname lecture "C:\PROC_PHREG";
data stan; set lecture.stan;
run;

/*Create the variables needed for analysis*/
data stan; set stan;
surv1 = dls - doa;
ageaccpt = (doa - dob)/365.25;
wait = dot - doa;
if dot = . then trans = 0;
else trans = 1;
keep surv1 dead wait surg ageaccpt trans;
run;
```

2. Wrong model: fit a Cox model for "trans" as a time-invariant covariate

```
/*Fit a cox model treat "trans" as a time-invariant variable*/
proc phreg data=stan;
model surv1*dead(0) = trans surg ageaccpt / ties=efron;
title "Cox model 1: with 'trans' as a time-invariant variable";
run;
```

Cox model 1: with 'trans' as a time-invariant variable

The PHREG Procedure

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio
trans	1	-1.44903	0.26363	30.2101	<.0001	0.235
surg	1	-0.00280	0.39767	0.0000	0.9944	0.997
ageaccpt	1	-0.01203	0.00317	14.3969	0.0001	0.988

1) Why this model is wrong?

- We define "trans" as a time-invariant covariate with a value of 1if the patient ever had a transplant and 0 otherwise, but patient's transplantation status can change with time during the follow-up.
- Patients who died quickly after acceptance into the program had less time available to get transplant.

3. Fit a Cox model for "trans" as a time-dependent covariate

1) Time-dependent covariates

- Time-dependent covariates are those that may change in value over the course of observation.
- We can modify a Cox model to include time-dependent covariates. For example, if we treat "trans" a time-dependent covariate, we have
  $h_i(t) = h_0(t)\exp\{\beta_1 surg_i + \beta_2 ageaccpt_i + \beta_3 trans_i(t)\}$.
  This says that the hazard at time t depends on the value of SURG and AGEACCPT as well as the value of TRANS at time t.

2) How to handle the time-dependent covariate in PROC PHREG?

- SAS syntax

```
/*Fit a cox model treat "trans" as a time-dependent variable*/
proc phreg data=stan;
model surv1*dead(0) = plant surg ageaccpt / ties=efron;
if wait >= surv1 or wait=. then plant = 0; else plant = 1;
title "Cox model 2: with 'trans' as a time-dependent variable";
run;
```

- The IF statement defines the new time-varying covariate PLANT. Note that programming statements must follow the MODEL statement. Unlike an IF statement in the DATA step, which only operates on a single case at a time, this IF statement compares waiting times for patients who were at risk of a death with survival times for patients who experienced events. Thus, SURV1 in the IF statement is not the patient's own survival time, but the survival times of all patients who had events.

Cox model 2: with 'trans' as a time-dependent variable

The PHREG Procedure

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio
plant	1	0.06426	0.30319	0.0449	0.8321	1.066
surg	1	-0.47084	0.37834	1.5488	0.2133	0.624
ageaccpt	1	-0.00848	0.00292	8.4180	0.0037	0.992

- Model 1 shows that the hazard for those who received a transplant is lower than that for those who did not. While Model 2 shows transplantation has no effect on the hazard of death.

3) For more examples of time-dependent covariates, see page 153-172 (Allison 2010).