

P8110 Applied Regression II

Midterm Project Report 1

Survival Analysis of Depression Onset and Substance Abuse in Offspring of Depressed Parents

Group 6

Leah Li (YL5828)
Xuange Liang (XL3493)
Zexuan Yan (ZY2654)
Yiwen Zhang (YZ4994)

November 5, 2025

Contents

1	Data Preparation	2
1.1	Survival Outcome Definitions	2
1.2	Data Cleaning	2
1.3	Descriptive Statistic Table	3
2	Descriptive Statistics Analyses	4
2.1	Sample Characteristics	4
2.2	Comparison by Parental Depression Status	4
3	Analyses to Address Hypothesis 1	5
3.1	Statistical Modeling Approach	5
3.2	Results	6
3.2.1	Pre-pubertal Onset (Age < 13 years)	6
3.2.2	Adolescent/Early Adulthood Onset (Age \geq 13 years)	7
3.3	Conclusion for Hypothesis 1	8
4	Analyses to Address Hypothesis 2	9
4.1	Statistical Modeling Approach	9
4.2	Results	10
4.2.1	Model 2a: Effect of Prior Depression	10
4.2.2	Model 2b: Effect of Parental Depression	11
4.2.3	Model 2c: Joint Model	12
4.3	Conclusion for Hypothesis 2	12
A	SAS Code for Statistical Analyses	13
A.1	Data Import, Preparation, and Descriptive Statistics	13
A.2	Hypothesis 1a: Pre-pubertal Depression Onset	14
A.3	Hypothesis 1b: Adolescent Depression Onset	15
A.4	Hypothesis 2: Substance Abuse Dataset Preparation	16
A.5	Hypothesis 2a: Effect of Prior Depression	16
A.6	Hypothesis 2b: Effect of Parental Depression	17
A.7	Hypothesis 2c: Joint Model	17

1 Data Preparation

1.1 Survival Outcome Definitions

Depression Onset (Pre-pubertal, age < 13):

- All 220 subjects included in analysis
- Start time: Birth (age = 0)
- Event: Depression onset before age 13 (BEDEPON < 13)
- Censoring: All others censored at age 13 (either had depression at/after 13, or never had depression)
- This creates a restricted follow-up period (0–13 years) to focus on pre-pubertal onset

Depression Onset (Adolescent/early adulthood, age ≥ 13):

- Includes only subjects who reached age 13 without depression (left-truncated at age 13)
- Excludes: (1) those with depression onset <13 years, and (2) those censored before age 13
- Start time: Age 13
- Event: Depression onset at age ≥13 (BEDEPON ≥ 13)
- Censoring: No depression by interview age (PTAGE)
- This is a conditional analysis among those “at risk” at age 13

Substance Abuse Onset:

- Start time: Birth (age = 0); End time: Age of substance abuse onset (BESUBON), censored at PTAGE if no event

1.2 Data Cleaning

Logical consistency checks were performed on all variables:

- Records with status=1 (diagnosed) but missing onset age (-1) were removed as incomplete
- Records with onset age present but status=0 were corrected to status=1 for consistency
- This procedure was applied to both depression (DSMDEPHR, BEDEPON) and substance abuse variables (DSMSUBHR, BESUBON)

Note: Complete SAS code for data preparation is provided in Appendix A.1.

1.3 Descriptive Statistic Table

Table 1: Demographic and Clinical Characteristics of the Offspring Stratified by Parental Depression Status

Variable	Depressed Index Parent Group (N=125) n (%)	Normal Index Parent Group (N=95) n (%)
Sex of Child		
Male	63 (50.4%)	42 (44.2%)
Female	62 (49.6%)	53 (55.8%)
Child depression status		
Ever depressed	47 (37.6%)	22 (23.2%)
Never depressed	78 (62.4%)	73 (76.8%)
Child substance abuse status		
Yes	21 (16.8%)	7 (7.4%)
No	104 (83.2%)	88 (92.6%)
Social class of index parent		
1 (Highest)	16 (12.8%)	1 (1.1%)
2	19 (15.2%)	19 (20.0%)
3	31 (24.8%)	16 (16.8%)
4	50 (40.0%)	47 (49.5%)
5 (Lowest)	8 (6.4%)	12 (12.6%)
Marital status of index parent		
Married with spouse	90 (72.0%)	84 (88.4%)
Separated/Divorced	35 (28.0%)	11 (11.6%)

Summary Statistics:

- Age at interview: Mean = 16.72 years (SD = 4.74, Range: 5–25)
- Age of depression onset (n=69): Mean = 15.19 years (SD = 4.74, Range: 4–25)
- Age of substance abuse onset (n=28): Mean = 21.46 years (SD = 3.53, Range: 12–25)

2 Descriptive Statistics Analyses

2.1 Sample Characteristics

The study included 220 children (125 with depressed parents, 95 with non-depressed parents) aged 5–25 years (Mean = 16.72, SD = 4.74). Depression onset occurred in 69 children (31.4%), with mean time to event of 15.19 years. Substance abuse was observed in 28 children (12.7%), with mean time to event of 21.46 years. Among those who developed depression, 24 cases (34.8%) had pre-pubertal onset (before age 13).

2.2 Comparison by Parental Depression Status

Table 1 shows demographic and clinical characteristics stratified by parental depression status. Children of depressed parents had higher rates of depression (37.6% vs 23.2%) and substance abuse (16.8% vs 7.4%). Separation/divorce was more common among depressed parents (28.0% vs 11.6%).

Overall Kaplan-Meier Analysis: Children of depressed parents demonstrated earlier depression onset (median survival: 20 vs. 23 years, log-rank $\chi^2 = 7.69$, $p = 0.0056$).




Figure 1: Overall Kaplan–Meier survival curves of depression onset by parental depression status

3 Analyses to Address Hypothesis 1

Hypothesis 1: Offspring of a depressed parent are more likely to have pre-pubertal onset (< 13 years) depression than offspring of a non-depressed parent, but equally likely to have an onset of adolescent/early adulthood depression, given demographic and social characteristics.

3.1 Statistical Modeling Approach

Cox proportional hazards regression models were fit separately for: (a) pre-pubertal onset (age < 13), and (b) adolescent/early adulthood onset (age ≥ 13).

Cox Model:

$$h(t|X) = h_0(t) \times \exp(\beta_1 \times \text{PARDEP} + \beta_2 \times \text{SEX} + \beta_3 \times \text{AGE} + \beta_4 \times \text{SESCLASS} + \beta_5 \times \text{MSPARENT}) \quad (1)$$

Where: $h(t|X)$ = hazard at time t ; $h_0(t)$ = baseline hazard; PARDEP = parental depression (1=depressed, 0=normal); SEX = child's sex (1=male, 2=female); AGE = age at interview; SESCLASS = social class (1–5); MSPARENT = marital status (1=married, 2=separated/divorced).

The proportional hazards (PH) assumption was assessed using the supremum test with 1000 replications, and results are reported for each hypothesis.

3.2 Results

3.2.1 Pre-pubertal Onset (Age < 13 years)

Sample: All 220 subjects included; 24 events (10.9%), 196 censored at age 13.

Kaplan-Meier (Figure 2): Log-rank $\chi^2 = 10.56$, $p = 0.0012$. The survival curves show a clear separation, with children of depressed parents experiencing significantly earlier pre-pubertal depression onset compared to children of non-depressed parents.




Figure 2: H1a: Kaplan–Meier curves for pre-pubertal depression onset (age < 13) by parental depression

Cox Regression (Table 2):

Table 2: Cox Regression Results for Pre-pubertal Depression Onset (Age < 13 years)

Variable	Hazard Ratio	95% CI	P-value
Parental Depression	5.863	(1.717, 20.019)	0.0048**
Sex (Female vs Male)	1.171	(0.522, 2.626)	0.7024
Age at interview	0.988	(0.913, 1.069)	0.7608
Social class	0.822	(0.591, 1.144)	0.2460
Marital status (Sep/Div)	0.585	(0.198, 1.731)	0.3328

** Statistically significant at $\alpha = 0.01$ level

Key Findings: Children of depressed parents have 5.86 times the hazard of pre-pubertal depression (HR=5.863, 95% CI: 1.717–20.019, $p=0.0048$). No other covariates were significant. The wide confidence interval reflects the relatively small number of pre-pubertal depression events ($n=24$).

Proportional Hazards Assessment: The PH assumption was assessed using the supremum test with 1000 replications. All covariates satisfied the PH assumption: parental depression ($p = 0.062$), gender ($p = 0.400$), age ($p = 0.184$), social class ($p = 0.090$), and marital status ($p = 0.228$). The

Cox model assumptions are met for all predictors in the pre-pubertal analysis.

Note: Complete SAS code for pre-pubertal analysis is provided in Appendix A.2.

3.2.2 Adolescent/Early Adulthood Onset (Age ≥ 13 years)

Sample: 154 subjects who reached age 13 without depression (excluded 66 subjects: 24 with pre-pubertal onset and 42 censored before age 13).

Kaplan-Meier (Figure 3): Log-rank $\chi^2 = 0.83$, $p = 0.3632$ (not significant). No difference by parental status, consistent with hypothesis.




Figure 3: H1b: Kaplan–Meier curves for adolescent/early adulthood depression onset (age ≥ 13) by parental depression

Cox Regression (Table 3):

Table 3: Cox Regression Results for Adolescent/Early Adulthood Depression Onset (Age ≥ 13 years)

Variable	Hazard Ratio	95% CI	P-value
Parental Depression	1.100	(0.567, 2.134)	0.7779
Sex (Female vs Male)	2.226	(1.125, 4.404)	0.0216*
Age at interview	0.979	(0.847, 1.131)	0.7695
Social class	1.139	(0.828, 1.566)	0.4240
Marital status (Sep/Div)	1.718	(0.824, 3.583)	0.1492

* Statistically significant at $\alpha = 0.05$ level

Key Findings: Parental depression not significant (HR=1.100, 95% CI: 0.567–2.134, $p=0.778$). Female sex predicts adolescent depression (HR=2.226, 95% CI: 1.125–4.404, $p=0.022$), indicating females have more than double the hazard of adolescent-onset depression compared to males.

Note on Sample Size: The reduction from 154 to 136 subjects (18 excluded due to missing covariates) may affect power but does not alter the substantive conclusion that parental depression is not associated with adolescent-onset depression.

Proportional Hazards Assessment: Due to limited events ($n=38$), the Supremum test could not be computed. Exploratory analyses suggested potential PH violations for age ($p=0.021$). A sensitivity analysis using age \times log(time) interaction showed non-significant interaction (HR=2.194, $p=0.257$), with the main finding unchanged (parental depression HR=1.115, $p=0.748$).

Note: Complete SAS code for adolescent analysis with left-truncation is provided in Appendix A.3.

3.3 Conclusion for Hypothesis 1

Hypothesis 1 is FULLY SUPPORTED. Both components of the hypothesis were confirmed:

(a) Pre-pubertal analysis: Parental depression significantly increased depression risk (HR=5.863, 95% CI: 1.717–20.019, $p=0.0048$), demonstrating that offspring of depressed parents are more likely to experience early-onset depression.

(b) Adolescent analysis: Parental depression showed no significant association (HR=1.100, 95% CI: 0.567–2.134, $p=0.778$), confirming that offspring are equally likely to have adolescent-onset depression regardless of parental depression status. Female sex was the only significant predictor for adolescent-onset (HR=2.226, 95% CI: 1.125–4.404, $p=0.022$). Time-interaction sensitivity analysis confirmed robustness of findings.

4 Analyses to Address Hypothesis 2

Hypothesis 2: Is there evidence for the effect of (1) prior depression in offspring, as well as (2) the effect of parent's depression status on the age of onset of substance abuse/dependence in offspring, given demographic and social characteristics?

4.1 Statistical Modeling Approach

Three Cox regression models tested: (2a) prior depression effect, (2b) parental depression effect, (2c) joint model.

Definition of “Prior Depression”: Prior depression is operationalized as having a history of depression during the observation period. For offspring who developed substance abuse, prior depression indicates depression onset preceded substance abuse onset ($BEDEPON < BESUBON$). For those who did not develop substance abuse, prior depression indicates any history of depression by the interview age. This definition captures whether offspring with depression history have different substance abuse risk profiles, treating depression as a baseline characteristic rather than a time-varying covariate.

Model Specifications:

$$\text{Model 2a: } h(t|X) = h_0(t) \times \exp(\beta_1 \times \text{PRIORDEP} + \text{covariates}) \quad (2)$$

$$\text{Model 2b: } h(t|X) = h_0(t) \times \exp(\beta_1 \times \text{PARDEP} + \text{covariates}) \quad (3)$$

$$\text{Model 2c: } h(t|X) = h_0(t) \times \exp(\beta_1 \times \text{PRIORDEP} + \beta_2 \times \text{PARDEP} + \text{covariates}) \quad (4)$$

Where PRIORDEP = prior depression (1=yes, 0=no); PARDEP = parental depression; covariates = sex, age, social class, marital status. The proportional hazards assumption was assessed for the joint model (Model 2c) using the same methods as Hypothesis 1.

4.2 Results

4.2.1 Model 2a: Effect of Prior Depression

Sample: 220 subjects with complete data; 28 substance abuse events (12.7%), 192 censored (87.3%).

Kaplan-Meier (Figure 4): Log-rank $\chi^2 = 0.07$, $p = 0.7988$ (not significant). No difference in substance abuse onset by prior depression status.




Figure 4: H2a: Kaplan–Meier curves for substance abuse onset by prior depression in offspring

Cox Regression (Table 4):

Table 4: Cox Regression Results for Substance Abuse Onset: Effect of Prior Depression

Variable	Hazard Ratio	95% CI	P-value
Prior Depression	0.999	(0.454, 2.198)	0.9981
Sex (Female vs Male)	0.538	(0.247, 1.170)	0.1155
Age at interview	1.045	(0.889, 1.228)	0.5981
Social class	1.000	(0.692, 1.445)	0.9989
Marital status (Sep/Div)	0.441	(0.149, 1.303)	0.1388

Key Findings: Prior depression shows no effect (HR=0.999, 95% CI: 0.454–2.198, $p=0.998$). No covariates significant.

4.2.2 Model 2b: Effect of Parental Depression

Kaplan-Meier (Figure 5): Log-rank $\chi^2 = 3.14$, $p = 0.0764$ (marginally significant). Trend toward earlier substance abuse in children of depressed parents.




Figure 5: H2b: Kaplan–Meier curves for substance abuse onset by parental depression

Cox Regression (Table 5):

Table 5: Cox Regression Results for Substance Abuse Onset: Effect of Parental Depression

Variable	Hazard Ratio	95% CI	P-value
Parental Depression	2.466	(1.001, 6.076)	0.0498*
Sex (Female vs Male)	0.592	(0.271, 1.293)	0.1794
Age at interview	1.021	(0.869, 1.200)	0.8018
Social class	1.107	(0.771, 1.591)	0.5728
Marital status (Sep/Div)	0.378	(0.128, 1.117)	0.0801

* Statistically significant at $\alpha = 0.05$ level

Key Findings: Parental depression marginally significant (HR=2.466, 95% CI: 1.001–6.076, $p=0.0498$). Children of depressed parents have approximately 2.5 times the hazard of substance abuse.

Proportional Hazards Assessment: The PH assumption was assessed using the Supremum test with 1000 replications. All covariates satisfied the PH assumption with no significant violations detected (all $p > 0.05$). The model assumptions are met for the parental depression analysis.

4.2.3 Model 2c: Joint Model

Cox Regression (Table 6):

Table 6: Cox Regression Results for Substance Abuse Onset: Joint Model (Prior and Parental Depression)

Variable	Hazard Ratio	95% CI	P-value
Prior Depression	0.819	(0.372, 1.802)	0.6273
Parental Depression	2.586	(1.023, 6.539)	0.0430*
Sex (Female vs Male)	0.605	(0.277, 1.322)	0.2008
Age at interview	1.024	(0.871, 1.203)	0.7809
Social class	1.112	(0.770, 1.606)	0.5529
Marital status (Sep/Div)	0.366	(0.121, 1.106)	0.0740

* Statistically significant at $\alpha = 0.05$ level

Key Findings: Parental depression remains significant (HR=2.586, 95% CI: 1.023–6.539, $p=0.043$) in joint model. Prior depression not significant (HR=0.819, 95% CI: 0.372–1.802, $p=0.627$). Parental depression is the primary driver of substance abuse risk, with children of depressed parents having approximately 2.6 times the hazard of substance abuse regardless of their own depression status.

4.3 Conclusion for Hypothesis 2

Hypothesis 2 is PARTIALLY SUPPORTED. Prior depression in offspring shows no association with substance abuse onset (Model 2a: HR=0.999, 95% CI: 0.454–2.198, $p=0.998$).

Parental depression demonstrates a marginally significant effect (Model 2b: HR=2.466, 95% CI: 1.001–6.076, $p=0.0498$).

The joint model (Model 2c) confirms that parental depression remains significant (HR=2.586, 95% CI: 1.023–6.539, $p=0.043$) while prior depression does not (HR=0.819, $p=0.627$), indicating that parental depression directly affects substance abuse risk independent of offspring's own depression history. All models satisfied PH assumptions (all $p_i > 0.05$).

Note: Complete SAS code for substance abuse analyses is provided in Appendix A.4–A.7.

A SAS Code for Statistical Analyses

This appendix contains the complete SAS code used for all analyses reported in this study. The code is organized by analysis section corresponding to the main report.

A.1 Data Import, Preparation, and Descriptive Statistics

```

1 %
2 /* Create analysis dataset with survival variables */
3 data survival_data;
4   set rawdata;
5
6   /* Depression survival variables */
7   if DSMDEPHR = 0 and BEDEPON = -1 then do;
8     censor = 1;          /* Censored (no depression) */
9     time = PTAGE;        /* Use age at interview */
10  end;
11  else if DSMDEPHR = 1 and BEDEPON ne -1 then do;
12    censor = 0;          /* Event occurred */
13    time = BEDEPON;      /* Use age of onset */
14  end;
15  else if DSMDEPHR = 1 and BEDEPON = -1 then do;
16    censor = 0;          /* Event occurred, onset unknown */
17    time = PTAGE;        /* Use interview age */
18  end;
19
20  /* Rename variables for clarity */
21  depress_parent = PARDEP;
22  depress_child = DSMDEPHR;
23  substance_abuse = DSMSUBHR;
24  social_class = SESCLASS;
25  marital_status = MSPARENT;
26  age = PTAGE;
27  gender = PTSEX;
28
29  /* Create labeled categories */
30  if gender = 1 then gender_label = "Male";
31  else if gender = 2 then gender_label = "Female";
32
33  if depress_parent = 0 then parent_label = "No Parental Depression";
34  else if depress_parent = 1 then parent_label = "Parental Depression";
35
36  /* Keep only valid observations */
37  if time > 0 and time ne .;
38
39  label
40    censor = "Censoring Status (1=Censored, 0=Event)"
41    time = "Time to Depression Onset or Censoring"
42    depress_parent = "Parental Depression (0=No, 1=Yes)"
43    social_class = "Social Class (1=Highest, 5=Lowest)"
44    marital_status = "Marital Status (1=Married, 2=Sep/Div)"
45    age = "Age at Interview"
46    gender = "Gender (1=Male, 2=Female)";
47 run;

```

Listing 1: Data import and preparation

```

1 /* Frequency tables by parental depression status */
2 proc freq data=survival_data;
3     tables (gender depress_child substance_abuse social_class marital_status)
4         * parent_label / norow nocol nopercnt;
5     title "Descriptive Statistics by Parental Depression";
6 run;
7
8 /* Summary statistics for continuous variables */
9 proc means data=survival_data n mean std min max;
10     var time age;
11     title "Summary Statistics";
12 run;
13
14 /* Overall Kaplan-Meier curves for depression onset */
15 ods graphics on;
16 proc lifetest data=survival_data method=KM plots=survival(test);
17     time time*censor(1);
18     strata parent_label;
19     title 'Overall KM Survival Curves: Depression Onset by Parental Status';
20 run;
21 ods graphics off;

```

Listing 2: Descriptive statistics

A.2 Hypothesis 1a: Pre-pubertal Depression Onset

```

1 /* Create dataset for pre-pubertal analysis (<13 years) */
2 data prepubertal_data;
3     set survival_data;
4
5     if censor = 0 and time < 13 then do;
6         time_prepub = time;          /* Event occurred before 13 */
7         censor_prepub = 0;          /* Event */
8     end;
9     else do;
10        time_prepub = 13;             /* Censored at age 13 */
11        censor_prepub = 1;           /* Censored */
12    end;
13 run;
14
15 /* Kaplan-Meier curves */
16 ods graphics on;
17 proc lifetest data=prepubertal_data method=KM plots=survival(test);
18     time time_prepub*censor_prepub(1);
19     strata parent_label;
20     title 'H1a: Pre-pubertal Depression Onset (<13 years)';
21 run;
22 ods graphics off;
23
24 /* Cox regression */
25 proc phreg data=prepubertal_data;
26     class depress_parent (ref='0') gender (ref='1') marital_status (ref='1');
27     model time_prepub*censor_prepub(1) = depress_parent gender age
28         social_class marital_status / rl;
29     title 'H1a: Cox Regression for Pre-pubertal Depression Onset';
30 run;
31

```

```

32 /* PH assumption test */
33 ods graphics on;
34 proc phreg data=prepubertal_data;
35     class depress_parent (ref='0') gender (ref='1') marital_status (ref='1');
36     model time_prepub*censor_prepub(1) = depress_parent gender age
37         social_class marital_status / ties=efron;
38     assess PH / resample;
39     title 'H1a: Proportional Hazards Assumption Test';
40 run;
41 ods graphics off;

```

Listing 3: H1a: Pre-pubertal onset analysis

A.3 Hypothesis 1b: Adolescent Depression Onset

```

1  /* Create dataset for adolescent analysis (>=13 years) */
2  data adolescent_data;
3      set survival_data;
4
5      /* Exclude those with depression before 13 or censored before 13 */
6      if time < 13 then delete;
7
8      /* Left truncation: all individuals enter at age 13 */
9      entry_time = 13;
10     exit_time = time;
11     event = (censor = 0);
12
13     /* For KM curves: time since age 13 */
14     time_since_13 = exit_time - 13;
15     censor_km = 1 - event;
16 run;
17
18 /* Kaplan-Meier curves */
19 ods graphics on;
20 proc lifetest data=adolescent_data method=KM plots=survival(test);
21     time time_since_13*censor_km(1);
22     strata parent_label;
23     title 'H1b: Adolescent/Early Adulthood Depression Onset (>=13 years)';
24 run;
25 ods graphics off;
26
27 /* Cox regression with left truncation */
28 proc phreg data=adolescent_data;
29     class depress_parent (ref='0') gender (ref='1') marital_status (ref='1');
30     model (entry_time, exit_time)*event(0) = depress_parent gender age
31         social_class marital_status / rl;
32     title 'H1b: Cox Regression for Adolescent/Early Adulthood Depression';
33 run;
34
35 /* Sensitivity analysis: age*time interaction */
36 proc phreg data=adolescent_data;
37     class depress_parent (ref='0') gender (ref='1') marital_status (ref='1');
38     model (entry_time, exit_time)*event(0) = depress_parent gender age
39         social_class marital_status age_time;
40     age_time = age * log(exit_time);
41     title 'H1b: Cox Model with Age*Time Interaction';
42 run;

```


Listing 4: H1b: Adolescent onset analysis with left truncation

A.4 Hypothesis 2: Substance Abuse Dataset Preparation

```

1  /* Create substance abuse analysis dataset */
2  data substance_analysis;
3      set rawdata;
4
5      /* Substance abuse time and censoring */
6      if DSMSUBHR = 1 and BESUBON ne -1 then do;
7          time_subst = BESUBON;
8          censor_subst = 0;
9      end;
10     else if DSMSUBHR = 1 and BESUBON = -1 then do;
11         time_subst = PTAGE;
12         censor_subst = 0;
13     end;
14     else if DSMSUBHR = 0 then do;
15         time_subst = PTAGE;
16         censor_subst = 1;
17     end;
18
19     /* Prior depression variable */
20     prior_depression = 0;
21     if DSMDEPHR = 1 and BEDEPON ne -1 then do;
22         if DSMSUBHR = 1 and BESUBON ne -1 then do;
23             if BEDEPON < BESUBON then prior_depression = 1;
24         end;
25         else if DSMSUBHR = 1 and BESUBON = -1 then prior_depression = 1;
26         else if DSMSUBHR = 0 then prior_depression = 1;
27     end;
28
29     /* Rename variables */
30     depress_parent = PARDEP;
31     gender = PTSEX;
32     age = PTAGE;
33     social_class = SESCLASS;
34     marital_status = MSPARENT;
35
36     if time_subst > 0 and time_subst ne .;
37 run;

```

Listing 5: H2: Substance abuse analysis dataset preparation

A.5 Hypothesis 2a: Effect of Prior Depression

```

1  /* Kaplan-Meier curves */
2  ods graphics on;
3  proc lifetest data=substance_analysis method=KM plots=survival(test);
4      time time_subst*censor_subst(1);
5      strata prior_depression;
6      title 'H2a: Substance Abuse Onset by Prior Depression in Offspring';
7  run;
8  ods graphics off;
9

```

```

10 /* Cox regression */
11 proc phreg data=substance_analysis;
12     class prior_depression (ref='0') gender (ref='1') marital_status (ref='1');
13     model time_subst*censor_subst(1) = prior_depression gender age
14         social_class marital_status;
15     title 'H2a: Cox Regression - Effect of Prior Depression';
16 run;

```

Listing 6: H2a: Effect of prior depression

A.6 Hypothesis 2b: Effect of Parental Depression

```

1 /* Kaplan-Meier curves */
2 ods graphics on;
3 proc lifetest data=substance_analysis method=KM plots=survival(test);
4     time time_subst*censor_subst(1);
5     strata depress_parent;
6     title 'H2b: Substance Abuse Onset by Parental Depression Status';
7 run;
8 ods graphics off;
9
10 /* Cox regression */
11 proc phreg data=substance_analysis;
12     class depress_parent (ref='0') gender (ref='1') marital_status (ref='1');
13     model time_subst*censor_subst(1) = depress_parent gender age
14         social_class marital_status;
15     title 'H2b: Cox Regression - Effect of Parental Depression';
16 run;
17
18 /* PH assumption test */
19 ods graphics on;
20 proc phreg data=substance_analysis;
21     class depress_parent (ref='0') gender (ref='1') marital_status (ref='1');
22     model time_subst*censor_subst(1) = depress_parent gender age
23         social_class marital_status / ties=efron;
24     assess PH / resample;
25     title 'H2b: Proportional Hazards Assumption Test';
26 run;
27 ods graphics off;

```

Listing 7: H2b: Effect of parental depression

A.7 Hypothesis 2c: Joint Model

```

1 /* Joint model with both prior and parental depression */
2 proc phreg data=substance_analysis;
3     class prior_depression (ref='0') depress_parent (ref='0')
4         gender (ref='1') marital_status (ref='1');
5     model time_subst*censor_subst(1) = prior_depression depress_parent
6         gender age social_class marital_status;
7     title 'H2c: Cox Regression - Joint Model (Prior and Parental Depression)';
8 run;

```

Listing 8: H2c: Joint model